
7.14 <search or cycle clause>

Function

Specify the generation of ordering and cycle detection information in the result of recursive query expressions.

Format

<search or cycle clause> ::=
<search clause>

  | <cycle clause>
  | <search clause> <cycle clause>

<search clause> ::=
SEARCH <recursive search order> SET <sequence column>

<recursive search order> ::=
DEPTH FIRST BY <sort specification list>

  | BREADTH FIRST BY <sort specification list>

<sequence column> ::= <column name>

<cycle clause> ::=
CYCLE <cycle column list> SET <cycle mark column> TO <cycle mark value>
DEFAULT <non-cycle mark value> USING <path column>

<cycle column list> ::=
<cycle column> [ { <comma> <cycle column> }... ]

<cycle column> ::= <column name>

<cycle mark column> ::= <column name>

<path column> ::= <column name>

<cycle mark value> ::= <value expression>

<non-cycle mark value> ::= <value expression>

Syntax Rules

1) Let WLEC be an expandable <with list element> immediately containing a <search or cycle clause>.

2) Let WQN be the <query name>, WCL the <with column list>, and WQE the <query expression> immediately
contained in WLEC. Let WQEB be the <query expression body> immediately contained in WQE. Let OP
be the set operator immediately contained in WQEB. Let TLO be the <query expression body> that constitutes
the first operand of OP and let TRO be the <query specification> that (necessarily) constitutes the second
operand of OP.

a) Let TROSL be the <select list> immediately contained in TRO. Let WQNTR be the <table reference>
simply contained in the <from clause> immediately contained in the <table expression> TROTE
immediately contained in TRO such that WQNTR immediately contains WQN.

